/*

	enviroment.prg
	Procesos graficos de decoracion del fondo
	
*/


GLOBAL
	int windVelocity = 2;	// velocidad del viento
END


PROCESS cloud(layer)
//
// Proceso cloud
// Nube con 3 niveles de profundidad
//
PRIVATE
	int animDelay;
END

BEGIN

	file = getAssetID("decorations");
	graph = rand(1,2);
	x = rand(100,500);
	
	// inicializa valores de acuerdo al layer
	SWITCH ( layer )
		CASE 1:	// MAS CERCANO
			y = rand(0, 50);
		END		
		CASE 2: // MEDIO
			y = rand(60, 120);
			size = 70;
		END
		CASE 3: // MAS LEJANO
			y = rand(130, 170);
			size = 40;
		END
	END
	
	
	LOOP
		animDelay++;
		
		IF ( animDelay > layer )
			animDelay = 0;
			x -= 1;
		END
		
		IF ( x < -200)
			// si desaparece por la izquierda lo pone en la derecha
			// con propiedades nuevas
			x = SCREEN_X + 200;
		END
		
	
		FRAME;
		
	END

END

PROCESS sun()
//
// Solcito que se mueve leeeento
//

PRIVATE
	int retraso;
END

BEGIN

	file = getAssetID("decorations");
	graph = 10;
	x = 100;
	y = 0;
	z = 10;

	LOOP
		retraso++;
		
		IF ( retraso > 60 )
			retraso = 0;
			x++;
		END
		
	
		frame;
		
	END

END
